

G08DAF – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users' Note for your implementation to check the interpretation of bold italicised terms and other implementation-dependent details.

1 Purpose

G08DAF calculates Kendall's coefficient of concordance on k independent rankings of n objects or individuals.

2 Specification

```
SUBROUTINE G08DAF(X, IX, K, N, RNK, W, P, IFAIL)
  INTEGER          IX, K, N, IFAIL
  real           X(IX,N), RNK(IX,N), W, P
```

3 Description

Kendall's coefficient of concordance measures the degree of agreement between k comparisons of n objects, the scores in the i th comparison being denoted by

$$x_{i1}, x_{i2}, \dots, x_{in}.$$

The hypothesis under test, H_0 , often called the null hypothesis, is that there is no agreement between the comparisons, and this is to be tested against the alternative hypothesis H_1 that there is some agreement.

The n scores for each comparison are ranked, the rank r_{ij} denoting the rank of object j in comparison i , and all ranks lying between 1 and n . Average ranks are assigned to tied scores.

For each of the n objects, the k ranks are totalled, giving rank sums R_j , for $j = 1, 2, \dots, n$. Under H_0 , all the R_j would be approximately equal to the average rank sum $k(n+1)/2$. The total squared deviation of the R_j 's from this average value is therefore a measure of the departure from H_0 exhibited by the data. If there were complete agreement between the comparisons, the rank sums R_j would have the values $k, 2k, \dots, nk$ (or some permutation thereof). The total squared deviation of these values is $k^2(n^3 - n)/12$.

Kendall's coefficient of concordance is the ratio

$$W = \frac{\sum_{j=1}^n (R_j - \frac{1}{2}k(n+1))^2}{\frac{1}{12}k^2(n^3 - n)}$$

and lies between 0 and 1, the value 0 indicating complete disagreement, and 1 indicating complete agreement.

If there are tied rankings within comparisons, W is corrected by subtracting $k \sum T$ from the denominator, where $T = \sum (t^3 - t)/12$, each t being the number of occurrences of each tied rank within a comparison, and the summation of T being over all comparisons containing ties.

G08DAF returns the value of W , and also an approximation, p , of the significance of the observed W . (For $n > 7$, $k(n-1)W$ approximately follows a χ_{n-1}^2 distribution, so large values of W imply rejection of H_0). H_0 is rejected by a test of chosen size α if $p < \alpha$. If $n \leq 7$, tables should be used to establish the significance of W (e.g., table R of Siegel [1]).

4 References

- [1] Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw-Hill

5 Parameters

- 1:** X(IX,N) — *real* array *Input*
On entry: X(i, j) must be set to the value x_{ij} of object j in comparison i , for $i = 1, 2, \dots, k$; $j = 1, 2, \dots, n$.
- 2:** IX — INTEGER *Input*
On entry: the first dimension of the array X as declared in the (sub)program from which G08DAF is called.
Constraint: IX \geq K.
- 3:** K — INTEGER *Input*
On entry: the number of comparisons, k .
Constraint: K \geq 2.
- 4:** N — INTEGER *Input*
On entry: the number of objects, n .
Constraint: N \geq 2.
- 5:** RNK(IX,N) — *real* array *Workspace*
- 6:** W — *real* *Output*
On exit: the value of Kendall's coefficient of concordance, W .
- 7:** P — *real* *Output*
On exit: the approximate significance, p , of W .
- 8:** IFAIL — INTEGER *Input/Output*
On entry: IFAIL must be set to 0, -1 or 1. For users not familiar with this parameter (described in Chapter P01) the recommended value is 0.
On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors detected by the routine:

IFAIL = 1

On entry, N < 2.

IFAIL = 2

On entry, IX < K.

IFAIL = 3

On entry, K \leq 1.

7 Accuracy

All computations are believed to be stable. The statistic W should be accurate enough for all practical uses.

8 Further Comments

The time taken by the routine is approximately proportional to the product nk .

9 Example

This example is taken from page 234 of Siegel [1]. The data consists of 10 objects ranked on three different variables: X, Y and Z. The computed values of Kendall's coefficient is significant at the 1% level of significance ($p = 0.008 < 0.01$), indicating that the null hypothesis of there being no agreement between the three rankings X, Y, Z may be rejected with reasonably high confidence.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details. Please read the Users' Note for your implementation to check the interpretation of these terms. As explained in the Essential Introduction to this manual, the results produced may not be identical for all implementations.

```

*      G08DAF Example Program Text
*      Mark 14 Revised.  NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          N, K, IX
      PARAMETER       (N=10,K=3,IX=K)
      INTEGER          NIN, NOUT
      PARAMETER       (NIN=5,NOUT=6)
*      .. Local Scalars ..
      real            P, W
      INTEGER          I, IFAIL, J
*      .. Local Arrays ..
      real            RNK(IX,N), X(IX,N)
*      .. External Subroutines ..
      EXTERNAL        G08DAF
*      .. Executable Statements ..
      WRITE (NOUT,*) 'G08DAF Example Program Results'
*      Skip heading in data file
      READ (NIN,*)
      READ (NIN,*) ((X(I,J),J=1,N),I=1,K)
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Kendall''s coefficient of concordance'
      WRITE (NOUT,*)
      WRITE (NOUT,*) 'Data values'
      WRITE (NOUT,*)
      WRITE (NOUT,99999) ('Comparison ',I,' scores ',(X(I,J),J=1,N),
+ I=1,K)
      IFAIL = 0
*
      CALL G08DAF(X,IX,K,N,RNK,W,P,IFAIL)
*
      WRITE (NOUT,*)
      WRITE (NOUT,99998) 'Kendall''s coefficient =', W
      WRITE (NOUT,99998) '          Significance =', P
      STOP
*
99999 FORMAT (1X,A,I1,A,10F5.1)
99998 FORMAT (1X,A,F8.3)
      END

```

9.2 Program Data

```

G08DAF Example Program Data
  1.0  4.5  2.0  4.5  3.0  7.5  6.0  9.0  7.5 10.0
  2.5  1.0  2.5  4.5  4.5  8.0  9.0  6.5 10.0  6.5
  2.0  1.0  4.5  4.5  4.5  4.5  8.0  8.0  8.0 10.0

```

9.3 Program Results

G08DAF Example Program Results

Kendall's coefficient of concordance

Data values

Comparison 1 scores	1.0	4.5	2.0	4.5	3.0	7.5	6.0	9.0	7.5	10.0
Comparison 2 scores	2.5	1.0	2.5	4.5	4.5	8.0	9.0	6.5	10.0	6.5
Comparison 3 scores	2.0	1.0	4.5	4.5	4.5	4.5	8.0	8.0	8.0	10.0

Kendall's coefficient = 0.828
Significance = 0.008
